Computation Structures

Procedures & Stacks Worksheet



PUSH(X): Push Reg[x] onto stack

```
ADDC(SP,4,SP)
ST(Rx,-4,SP)
```

POP(X): Pop value at top of stack into Reg[x]

```
LD(SP,-4,RX)
SUBC(SP,4,SP)
```

ALLOCATE(k): Reserve k words of stack

```
ADDC(SP,4*k,SP)
```

DEALLOCATE(k): Release k words of stack

```
SUBC(SP,4*k,SP)
```

Stack discipline: leave stack the way you found it => for every PUSH(), there's a corresponding POP() or DEALLOCATE()

Activation record layout on the stack (aka stack frame):



CALLING SEQUENCE

```
   PUSH(argn)    // push args, last arg first
   …
   PUSH(arg1)
   BR(f, LP)     // call f, return addr in LP
   DEALLOCATE(n) // remove args from stack
```

ENTRY SEQUENCE

```
f: PUSH(LP)      // save return addr
   PUSH(BP)      // save old frame pointer
   MOVE(SP,BP)   // initialize new frame pointer
   ALLOCATE(nlocals)  // make room for locals
   (push other regs)  // preserve old reg vals
```

EXIT SEQUENCE

```
   // return value in R0
   MOVE(BP,SP)   // remove locals
   POP(BP)       // restore old frame pointer
   POP(LP)       // recover return address
   JMP(LP)       // resume execution in caller
```

Problem 1.

You are given an incomplete listing of a C program (shown below) and its translation to Beta assembly code (shown on the right):

```
int fn(int x) {
  int lowbit = x & 1;
  int rest = x >> 1;
  if (x == 0) return 0;
  else return ???;
}
```

```
fn: PUSH(LP)
    PUSH(BP)
    MOVE(SP,BP)
    ALLOCATE(2)
    PUSH(R1)
    LD(BP,-12,R0)
    ANDC(R0,1,R1)
xx: ST(R1,0,BP)
    SHRC(R0,1,R1)
    ST(R1,4,BP)
yy: BEQ(R0,rtn)
    LD(BP,4,R1)
    PUSH(R1)
    BR(fn,LP)
    DEALLOCATE(1)
    LD(BP,0,R1)
    ADD(R1,R0,R0)
rtn:POP(R1)
zz: MOVE(BP,SP)
    POP(BP)
    POP(LP)
    JMP(LP)
```

(A) What is the missing C source corresponding to ??? in the above program?

C source code: fn(rest) + lowbit

(B) Suppose the instruction bearing the tag '**zz:**' were eliminated from the assembly language program. Would the modified procedure work the same as the original procedure (circle one)?

Work the same? YES ... (NO)

The MOVE(BP,SP) is deallocating the local variables lowbit and rest

(C) In the space below, fill in the binary representation for the instruction stored at the location tagged '**xx:**' in the above program.

R27

0 1 1 0 0 1	0 0 0 0 1	1 1 0 1 1	0 0 0 0 0 0 0 0 0 0 0 0 0 0 0 0
ST	R1	BP	0

(fill in missing 1s and 0s for instruction at xx:)

The procedure **fn** is called from an external procedure and its execution is interrupted just prior to the execution of the instruction tagged '**yy:**'. The contents of a region of memory are shown on the left below.

NB: All addresses and data values are shown in hex. The contents of **BP** are 0x1C8 and **SP** contains 0x1D4.

(D) What was the argument to the most recent call to **fn**?

```
184:   4
188:   7
18C:  47   X
190:  C4   saved LP
194: 170   saved BP
198:   1   lowbit
19C:  23   rest
1A0:  22   saved R1
1A4:  23   X
1A8:  4C   saved LP
1AC: 198   saved BP
1B0:   1   lowbit
1B4:  11   rest
1B8:  23   saved R1
1BC:  11   X
1C0:  4C   saved LP
1C4: 1B0   saved BP
1C8:   1   ←BP  lowbit
1CC:   8   rest
1D0: ???   saved R1
1D4:   0   ←SP
```

from current stack frame **Most recent argument (HEX): x=** 11

(E) What is the missing value marked **???** for the contents of location 1D0?

looking at code: R1 held value of argument before the call

Contents of 1D0 (HEX): 11

(F) What is the hex address of the instruction tagged **rtn:?**

saved LP is the address of mem location holding DEALLOCATE inst.

Address of rtn (HEX): 58

(G) What was the argument to the *original* call to **fn**?

find frame with saved LP ≠ 0x4C.

Original argument (HEX): x= 47

(H) What is the hex address of the BR instruction that called **fn** *originally*?

(saved LP on oldest stack frame) − 4.

Address of original call (HEX): C0

(I) What were the contents of R1 at the time of the *original* call?

Original R1 contents (HEX): 22

(J) What value will be returned to the *original* caller?



Return value for original call (HEX): 4

fn counts number of 1 bits in argument.

Problem 2.

You are given an incomplete listing of a C program (shown below) and its translation to Beta assembly code (shown on the right):

```
int f(int x, int y) {
  x = (x >> 1) + y;
  if (y == 0) return x;
  else return ???;
}
```




```
f:    PUSH(LP)
      PUSH(BP)
      MOVE(SP,BP)
      PUSH(R1)
      LD(BP,-12,R0)
      SHRC(R0,1,R0)
      LD(BP,-16,R1)
      ADD(R0,R1,R0)
      BEQ(R1,rtn)
      SUBC(R1,1,R1)
      PUSH(R1)
      PUSH(R0)
      BR(f,LP)
      DEALLOCATE(2)
rtn:  POP(R1)
zz:   MOVE(BP,SP)
      POP(BP)
      POP(LP)
      JMP(LP)
```

(A) What is the missing C source corresponding to ??? in the above program?

C source code: f(x, y-1)

(B) Suppose the instruction bearing the tag '**zz:**' were eliminated from the assembly language program. Would the modified procedure work the same as the original procedure?

Work the same (circle one)? (YES) ... NO

The procedure **f** is called from an external procedure and then execution is stopped just prior to one of the executions of the instruction labeled '**rtn:**'. The addresses and contents of a region of memory are shown in the table on the right; all addresses and data values in the table are in hex. When execution is stopped **BP contains the value 0x14C and SP contains the value 0x150.**

(C) What are the arguments to the **currently active call** to **f**?

Most recent arguments (in hex): x = 0x 6 , y = 0x 1

(D) If you can tell from the information provided, specify the arguments to the **original** call to **f**, otherwise select **CAN'T TELL.**

Original arguments (in hex) : x = 0x A , y = 0x 3 , **or CAN'T TELL**

(E) What is the missing value in location 0x12C?

x = (0xA >> 1) + y
  = 5   3

Contents of location 0x12C (in hex): 0x 8

(F) What is the hex address of the instruction labeled **rtn:**?

Address of instruction labeled rtn: (in hex): 0x 34C

(G) What is the hex address of the BR instruction that called **f** *originally*?

Address of original call (in hex): 0x 2C0 , **or CAN'T TELL**

(H) What value will be returned to the *original* caller?

Return value for original call (in hex): 0x 2

f(A,3) → f(8,2) → f(6,1) → f(9,0) → returns 2

108	7	
10C	320	
110	104	
114	3	y
118	A	x
11C	2C4	LP
120	104	BP
124	3	R1
128	2	y
12C	8	x
130	348	LP
134	124	BP
138	2	R1
13C	1	y
140	6	x
144	348	LP
148	138	BP
BP→14C	1	R1
SP→150	0	
154	4	
158	348	
15C	14C	
160	0	

Problem 3.

The following C program implements a function H(x,y) of two arguments, which returns an integer result. The assembly code for the procedure is shown on the right.

```
int H(int x, int y) {
    int a = x - y;
    if (a < 0) return x;
    else return ???;
}
```




```
H:    PUSH(LP)
      PUSH(BP)
      MOVE(SP, BP)
      ALLOCATE(1)
      PUSH(R1)

      LD(BP,-12,R0)
      LD(BP,-16,R1)
      SUB(R0,R1,R1)
      ST(R1,0,BP)

      CMPLTC(R1,0,R1)
      BT(R1,rtn)

      LD(BP,-16,R1)
      PUSH(R1)
      LD(BP,0,R0)
      PUSH(R0)
      BR(H,LP)
      DEALLOCATE(2)

rtn:  POP(R1)
      MOVE(BP,SP)
      POP(BP)
      POP(LP)
      JMP(LP)
```

The execution of the procedure call **H(0x68,0x20)** has been suspended just as the Beta is about to execute the instruction labeled "rtn:" during one of the recursive calls to H. A *partial* trace of the stack at the time execution was suspended is shown to the right below.

(A) Examining the assembly language for H, what is the appropriate C code for ??? in the C representation for H?

C code for ???: H(a,y)

(B) Please fill in the values for the blank locations in the stack dump shown on the right. Express the values in hex or write "---" if value can't be determined. Hint: Figure out the layout of H's activation record and use it to identify and label the stack frames in the stack dump.

Fill in the blank locations with values (in hex!) or "---"

	0x0024	LP	
	0x0070	BP	
	0x0048	a	
	0x0068	R1	
	0x20	y	
	0x68	x	
	0x7C	LP	
	0x80	BP	
C8	0x28	a	
CC	0x0020	R1	
D0	0x0020	y	
D4	0x0028	x	
D8	0x007C	LP	
DC	0x00C8	BP	
E0	BP→ 0x0008	a	
E4	0x0020	R1	
SP→ E8	0x0020		

(C) Determine the specified values at the time execution was suspended. Please express each value in hex or write "CAN'T TELL" if the value cannot be determined.

result of H(8,20) **Value in R0 or "CANT TELL": 0x** 8

y **Value in R1 or "CANT TELL": 0x** 20

Value in BP or "CANT TELL": 0x E0

Value in LP or "CANT TELL": 0x 7C

Value in SP or "CANT TELL": 0x E8

Problem 4.

The following C program computes the log base 2 of its argument. The assembly code for the procedure is shown on the right, along with a stack trace showing the execution of ilog2(10). The execution has been halted just as it's about to execute the instruction labeled "rtn:"

```
/* compute log base 2 of arg */
int ilog2(unsigned x) {
    unsigned y;
    if (x == 0) return 0;
    else {
        /* shift x right by 1 bit */
        y = x >> 1;
        return ilog2(y) + 1;
    }
}
```


```
ilog2: PUSH(LP)
       PUSH(BP)
       MOVE(SP,BP)
       ALLOCATE(1)
       PUSH(R1)

       LD(BP,-12,R0)
       BEQ(R0,rtn,R31)

       LD(BP,-12,R1)
       SHRC(R1,1,R1)
       ST(R1,0,BP)

       LD(BP,0,R1)
       PUSH(R1)
       BR(ilog2,LP)
1A8    DEALLOCATE(1)
1AC    ADDC(R0,1,R0)
1B0,1B4
rtn:   POP(R1)        <-- STOP
1B8 xxx: DEALLOCATE(1)
       MOVE(BP,SP)
       POP(BP)
       POP(LP)
       JMP(LP)
```

(A) What are the values in R0, SP, BP and LP at the time execution was halted? Please express the values in hex or write "CAN'T TELL".

Value in R0: 0x 1 = ilog2(0)+1 **in SP: 0x** 24C

Value in BP: 0x 244 **in LP: 0x** 1A8

(B) Please fill in the values for the five blank locations in the stack trace shown on the right. Please express the values in hex.

Fill in values (in hex!) for 5 blank locations

(C) In the assembly language code for ilog2 there is the instruction "LD(BP,-12,R0)". If this instruction were rewritten as "LD(SP,NNN,R0)" what is correct value to use for NNN?

Correct value for NNN: -20

(D) In the assembly language code for ilog2, what is the address of the memory location labeled "xxx:"? Please express the value in hex.

Address of location labeled "xxx:": 0x 1B8

Values are in hex!

	5	x
	1A8	LP
	208	BP
	2	y
	5	R1
	2	x
	1A8	LP
	21C	BP
230	1	y
234	2	R1
238	1	x
23C	1A8	LP
240	230	BP
244 BP→	0	y
248	1	R1
24C SP→	0	

6.004 Computation Structures
Spring 2017